Labor of Fundamentals of Electronics and Power Electronics

Exercise No. 21

SIMMULATION OF LINEAR VOLTAGE REGULATORS

21.1 General information

The purpose of this exercise is simulation and test two circuits: simple voltage regulator with Zener diode and series voltage regulator with feed beck control loop of the output voltage. The PSpice 9.1 Student version is the simulator used in this exercise.

References:

Titze U., Schenk Ch.: Semiconductor Devices

PSpice Tutorial - http://dave.uta.edu/dillon/pspice/index.php

Baranecki A.: Laboratorium układów elektronicznych. Cz. 1

Kaźmierkowski M.P., Matysik J. T.: Wprowadzenie do elektroniki i energoelektroniki

21.2 Description of exercise

There are two files prepared to analyze (STAB1.CIR, STAB2.CIR)

The file STAB1.CIR containing the simple voltage regulator with Zener diode model. The resistance R1 are restrict the maximal value of Zener diode current for R2→∞.

The file STAB2.CIR containing the simple series voltage regulator with feed beck control loop of the output voltage model.

```
STAB1
**********************
*** STABILIZATOR 1 ***
**********************
R1 1 2 20
R2 2 0 MFR 50
**********************
**********
**********************
.TRAN .1ms 40ms
**********************
**********
**********************
*.DC RES MFR(R) 1 5 .1
**********************
**********
**********************
*.DC V1 1 15 .2
**********************
**********
**********
**********
**********
**********
**********
**********
D1 0 2 DZ7V5A
ra 1 0 1k
V1 1 0 sin(12 .5 50)
.PROBE
.MODEL MFR RES
* DZ4V model created using Parts version 1.03 on 06/17/91 at 08:43
*
*.model DZ4V D(Is=10f Rs=0 N=1 Xti=3 Eg=1.11 Bv=4.329 Ibv=12.93m Cjo=1p Vj=.75
*+  M=.3333 Fc=.5 Tt=5n)
*.model DZ4V7 D(Is=880.5E-18 Rs=.25 Ikf=0 N=1 Xti=3 Eg=1.11 Cjo=175p M=.5516
*+             Vj=.75 Fc=.5 Isr=1.859n Nr=2 Bv=4.7 Ibv=20.245m Nbv=1.6989
*+             Ibvl=1.9556m Nbvl=14.976 Tbvl=-21.277u)
* DZ7V5A model created using Parts version 5.0 on 03/14/92 at 00:31
*
.model DZ7V5A   D(Is=10f N=1 Rs=.1 Ikf=0 Xti=3 Eg=1.11 Cjo=1p M=.3333 Vj=.75
+               Fc=.5 Isr=100p Nr=2 Bv=7.447 Ibv=35.19m Tt=5n)

* DZ7V5 model created using Parts version 5.0 on 03/16/92 at 00:52
*
*.model DZ7V5    D(Is=10f N=1 Rs=.1 Ikf=0 Xti=3 Eg=1.11 Cjo=220.24p M=.3333
*+               Vj=.75 Fc=.5 Isr=100p Nr=2 Bv=7.447 Ibv=35.19m Tt=50n)
.end
```

The series voltage regulator with feed beck control loop of the output voltage model include two transistors Q1, Q2, Q1 is 2N3055 and Q2 is 2N2222. To simulations is possible to apply three models of transistor Q1 and three models of transistor Q2.

For Q1 is:

Q2N3055A, current gain $\beta = 20$

Q2N3055B, current gain $\beta = 100$,

Q2N3055C, current gain $\beta = 200$.

For Q2 is:

Q2N2222A, current gain $\beta = 100$

Q2N2222B, current gain $\beta = 500$

Q2N2222C, current gain $\beta = 1000$

The change of the transistor tape is possible after to register in the line instruction programme capital letter A, B or C.

For example Q1 1 2 3 Q2N3055A.

```
STAB2
**********************
*** STABILIZATOR 2 ***
**********************
R3 3 4 6.2k
R5 3 0 MFR 50
Q1 1 2 3 Q2N3055b
Q2 2 4 5 Q2N2222b
*********************
********************
*******************
******************
*****************
***********************
*.DC RES MFR(R) 1 10 .5
***********************
*************
************
***********
**********
*********
********
*******
******
*****
V1 1 0 sin(18 1 50)
R1 1 5 390
R2 1 2 150
R4 4 0 5k
D1 0 5 DZ4V3
R9 1 0 1k
*.DC V1 1 20 .2
.PROBE
.MODEL MFR RES
.TRAN .1ms 40ms
```

```
*.model DZ1A D(Is=10f Rs=0 N=1 Xti=3 Eg=1.11 Bv=4.329 Ibv=12.93m Cjo=1p Vj=.7
*+            M=.3333 Fc=.5 Tt=5n)
.model DZ4V3 D(Is=10f N=1 Rs=.1 Ikf=0 Xti=3 Eg=1.11 Cjo=1p M=.333 Vj=.7
+             Fc=.5 Isr=100p Nr=2 Bv=4.254 Ibv=10.37m Tt=5n)
.model Q2N2222A NPN(Is=3.1f Xti=3 Eg=1.11 Vaf=130 Bf=100 Ne=1.54 Ise=191f
+                Ikf=1.3 Xtb=1.5 Br=6.18 Nc=2 Isc=0 Ikr=0 Rc=1 Cjc=14.6p Vjc=.7
+                Mjc=.333 Fc=.5 Cje=26p Vje=.7 Mje=.333 Tr=51n Tf=451p
+                Itf=.1 Vtf=10 Xtf=2 Rb=10)
.model Q2N2222B NPN(Is=3.1f Xti=3 Eg=1.11 Vaf=130 Bf=500 Ne=1.54 Ise=191f
+                Ikf=1.3 Xtb=1.5 Br=6.18 Nc=2 Isc=0 Ikr=0 Rc=1 Cjc=14.6p Vjc=.7
+                Mjc=.333 Fc=.5 Cje=26p Vje=.7 Mje=.333 Tr=51n Tf=451p
+                Itf=.1 Vtf=10 Xtf=2 Rb=10)
.model Q2N2222C NPN(Is=3.1f Xti=3 Eg=1.11 Vaf=130 Bf=1000 Ne=1.54 Ise=191f
+                Ikf=1.3 Xtb=1.5 Br=6.18 Nc=2 Isc=0 Ikr=0 Rc=1 Cjc=14.6p Vjc=.7
+                Mjc=.333 Fc=.5 Cje=26p Vje=.7 Mje=.333 Tr=51n Tf=451p
+                Itf=.1 Vtf=10 Xtf=2 Rb=10)
.model Q2N3055A NPN(Is=975f Xti=3 Eg=1.11 Vaf=50 Bf=20 Ne=1.94 Ise=900p
+                Ikf=4 Xtb=1.5 Br=2.95 Nc=2 Isc=0 Ikr=0 Rc=.1 Cjc=276p Vjc=.75
+                Mjc=.333 Fc=.5 Cje=570p Vje=.75 Mje=.333 Tr=970n Tf=39n
+                Itf=20 Vtf=10 Xtf=2 Rb=10)
.model Q2N3055B NPN(Is=975f Xti=3 Eg=1.11 Vaf=50 Bf=100 Ne=1.94 Ise=900p
+                Ikf=4 Xtb=1.5 Br=2.95 Nc=2 Isc=0 Ikr=0 Rc=.1 Cjc=276p Vjc=.75
+                Mjc=.333 Fc=.5 Cje=570p Vje=.75 Mje=.333 Tr=970n Tf=39n
+                Itf=20 Vtf=10 Xtf=2 Rb=10)
.model Q2N3055C NPN(Is=975f Xti=3 Eg=1.11 Vaf=50 Bf=200 Ne=1.94 Ise=900p
+                Ikf=4 Xtb=1.5 Br=2.95 Nc=2 Isc=0 Ikr=0 Rc=.1 Cjc=276p Vjc=.75
+                Mjc=.333 Fc=.5 Cje=570p Vje=.75 Mje=.333 Tr=970n Tf=39n
+                Itf=20 Vtf=10 Xtf=2 Rb=10)
.options reltol = 100.000n trtol = 1 vntol = 10.000n  ; *ipsp*
.END
```

21.3 Schedule of exercise

21.3.1 Simulation of voltage regulator with Zener diode model. File STAB1.CIR. Fig 21.1.
Input voltage V1: 12 V DC, 0,5V/ 50 Hz AC.



Fig.21.1



Fig.21.2

a) For resistance R1 = 20 Ω and values of resistance R2: 50 Ω, 40 Ω, 30 Ω, 20 Ω measure the average value and maximum value of AC component in voltage V2.
b) Set up R2 = 20 Ω in model, disable DC analysis by putting the symbol "*" at the beginning of the line „**.DC V1 1 15 .2"**. and enable DC analysis by deleting symbol "*" at the beginning of the line „**.DC RES MFR(R) 1 5 .1**". This analysis (DC sweep) allows the observation of changes of DC components in current and voltage in function of change of resistance R2, form nominal value up to five times greater. In this case axis X is scaled in multiplier of the initial value of resistance R2. Make the simulation and observe the waveforms: **V2, I(D1), I(D1)*V(2), I(R1)*V(1,2)**. Analyse obtained results of the simulation.
c) Set up R2 = 50 Ω in model, disable DC analysis by putting the symbol "*" at the beginning of the line „**.DC RES MFR(R) 1 5 .1**". and enable DC analysis by deleting symbol "*" at the beginning of the line „**.DC V1 1 15 .2"**. This analysis allows the observation of changes of DC components in current and voltage in function of input voltage from 1 V up to 15 V with step 0.2V. Make the simulation and observe the waveforms: **V2, I(D1), I(R1)*V(1,2)**. Analyse obtained results of the simulation.

21.3.2. Simulation study of shunt DC voltage stabilizer with amplifier of control error realized by using a single-stage amplifier with bipolar transistor (file STAB2.CIR). Zener diode breakdown voltage U_{Z0D1} = 4.3 V. The supply voltage V1: DC component is 18 V, the amplitude of sine wave of AC component is 1 V and frequency is 50 Hz. Diagram is shown in Figure 21.2.

a) For resistance R5 = 50 Ω select the two transistors Q1, Q2 from group B and set up value of resistance R3 so that the value of DC component in the voltage V3 = 12 V. Measure the amplitude of AC component in voltage V3.
b) For transistor Q2 from group B analyse the current gain β of transistor Q1 effect on the amplitude of AC component in voltage V3.
c) For transistor Q1 from group B analyse the current gain β of transistor Q2 effect on the amplitude of AC component in voltage V3.
d) Select both transistors Q1, Q2 from group B, enable DC analysis by deleting symbol "*" at the beginning of the line „***.DC RES MFR(R) 1 5 .1**", set up resistance R3 = 20 Ω. Analyse change of resistance effect on the DC and AC component in voltage V3 and power loss in transistor Q1.